

LISTING CONTROLE PROGRAMMA 2

Alfred Debels

Scanned, ocr'ed and converted to PDF by HansO, 2001

DEZE LISTINGTESTER IS VAN TOEPASSING OP DE LISTINGS VANAF MSX-GIDS NR. 3

De listingtester is een hulpmiddel voor de lezers die de programma's uit ons blad zelf intikken. Achter de programma's, die met deze tester gecontroleerd kunnen worden, staat een controletelling. Door nu zo'n programma met deze tester te controleren wordt eveneens een telling verkregen die -per regel- overeen moet komen met de telling in het blad. Zijn de getallen achter de regelnummers hetzelfde, dan is de kans erg groot dat deze regel goed is overgenomen. Zeker is dit echter niet, want wanneer bijvoorbeeld twee letters in een regel worden omgewisseld dan is de telling wel goed, maar kan het programma toch fouten opleveren.

WERKWIJZE

Eerst moet dit controleprogramma FOUTLOOS overgenomen worden en worden weggeschreven op cassette of diskette als ASCII-file met de volgende opdrachten: SAVE"CAS:TESTER", A (voor cassette). SAVE"A:TESTER",A (voor diskette). Dit hoeft slechts éénmaal te gebeuren; verder kan dit programma voor alle listings worden gebruikt.

Vervolgens moet het te controleren programma worden ingetikt. Is dit gedaan, dan het programma eerst wegschrijven naar cassette of diskette en niet eerst RUNen, want bij sommige fouten kan het programma blijven hangen en bent u het kwijt; alles moet dan opnieuw worden ingetikt. Nu kan het controleprogramma worden ingelezen en achter het te controleren programma worden geplaatst met de opdracht:

MERGE"CAS:TESTER" (voor cassette).

MERGE"A:TESTER" (voor diskette).

Vervolgens kan de controle worden gestart met: RUN 65000.

Let erop de fouten te corrigeren in het originele programma. Na het laden van het controleprogramma staat dit namelijk achter het eigenlijke programma en vormt nu één geheel. In sommige gevallen kan dit extra stuk programma problemen geven b.v. met de geheugenruimte!

REM-regels worden niet gecontroleerd. Mocht het voorkomen dat REM-regels wel exact moeten worden overgenomen, dan zal dit bij het desbetreffende programma

worden vermeld. Bij kleine of eenvoudige programma's wordt geen controletelling geplaatst. Ook worden meestal geen tellingen geplaatst bij programma's die deel uitmaken van een artikel. Deze programma's (of stukken programma) verschijnen ook meestal niet op de cassette of diskette van de programma-service.

Sommige programma's hebben een eigen controle. Bij deze programma's ontbreekt de controletelling of staat slechts een telling voor het gedeelte dat niet door het programma zelf wordt gecontroleerd. (Meestal BASIC-gedeeltes bij een machinetaal programma).

Er staat geen 'zelftest' meer bij dit controleprogramma daar er afwijkingen in de telling kunnen voorkomen wanneer de controle via het beeldscherm plaatsvindt of wanneer het programma al een keer is geRUND. Om dezelfde reden dient het te controleren programma ook alleen ingeladen te worden voor de controle en moet men NIET eerst RUN geven. Na het RUNen zal de controle-bij sommige regels-een andere waarde geven!!

Let goed op de DAT A-regels. Hier worden de meeste fouten gemaakt. Bij het omwisselen van getallen in deze DATAregels wordt wel een juiste telling gegeven, maar het programma zal niet (of onjuist) werken!

```

65000 ' *****
65010 ' * LISTING TESTER 2 *
65020 ' * Door Alfred Debels *
65030 ' * (c)1986 De MSX Gids *
65040 ' * Amsterdam *
65050 ' *****
65060 '
65070 CLS: CLEAR: SCREEN 0: WIDTH 40: PRINT
65080 PRINT "Weergave via beeldscherm
of printer?"
65090 PRINT : PRINT "B of P?: "
65100 X$=INKEY$: IF X$<>"B" AND X$<>"P
" THEN 65100
65110 START=32769!
65120 RG=START+2: T=0
65130 REM
65140 VR=PEEK(START)+256*PEEK(START+1
)
65150 NR=PEEK(RG)+256*PEEK(RG+1)
65160 IF VR=0 THEN 65270
65170 IF NR>=65000! THEN 65270
65180 S=0: FOR I=RG+2 TO VR-1
65190 A=PEEK(I)
65200 IF A=143 THEN I=VR: GOTO 65220
65210 S= S+A: NEXT
65220 S=S MOD 256
65230 T=T+S
65240 IF X$="B" THEN PRINT USING "Reg
el: ##### - ###"; NR, S: GOTO 65260
65250 LPRINT USING "Regel: ##### - ###
"; NR, S
65260 START=VR: RG=VR+2: GOTO 65140
65270 IF X$="P" THEN 65290
65280 PRINT: PRINT "Totaal: "; T: END
65290 LPRINT "Totaal: "; T: END

```